start,image<
;slip<
_b a20,b20,c20,d20,e20
[
    Nimbi:

    Beregning af trækningstabeller.
    Beregning af lovlige træk.
    Coreversion

    Timing:

                KB=0        KB=1
    nimbi5:     47.31s      46.49s
    nimbi6:     43.44s      42.66s      core
    nimbi7:     36.47s      35.87s      buffer
]
i=10
d1:  qq                    ; tabela
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq

     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq

     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq
     qq  0     ,
d2:  qq  0     ,           ; lovlige
    i=i+100
d4:  qq  1.28
     qq  1.29  ,
d5:  qq  1.39
     qq  1.30
     qq  1.35  ,
d6:  qq  1.33
     qq  1.34
     qq  1.38
     qq  1.36  ,
d7:  qq  1.32
     qq  1.31
     qq  1.37  ,
d8:  qq  d4
     qq  d5
     qq  d6
     qqf d7
;
;       _p_r_o_c_e_d_u_r_e findes
;       qq tabel addr           s1
;       qq addr m               s2
;       return fundet           s3
;       return ej fundet        s4
;
_b b10
c1:  arn(s2)   , mb rb1    ; R:=m; R:=R ∧ _3_5 0 _5 m;
     ck  -10   , ga rb2    ; R:=R _s_h_i_f_t -10; rb2 addr := Raddr
     arn(s2)   , ck  -15   ; R:=m; Raddr := R _s_h_i_f_t -5;
     ar  s1    , ga  rb3   ; Raddr:=Raddr+tabel addr; rb3:=Raddr;
b3:  arn  _0                ; R:=tabel[m ∧ _3_5 0 _5 m];
b2:  ns  _0     , ck s-1    ; R:=R _s_h_i_f_t (-(R ∧ _3_5 0 _5 m)-1);
     hr  s3            LO  ; _i_f R<0 _t_h_e_n return s3
     hr  s4                ; _e_l_s_e return s4;
b1:  qq 31.39              ; _3_5 0 _5 m
_e
;
;       _p_r_o_c_e_d_u_r_e indsæt1
;       qq tabeladdr            s1
;       qq addr m               s2
;
_b b10
c5:  arn(s2)   , mb  rb1   ; R:=m; R:=R ∧ _3_5 0 _5 m;
     ck  -10   , ga  rb2   ; R:=R _s_h_i_f_t -10; rb2 := Raddr;
     arn(s2)   , ck  -15   ; R:=m; R:=R _s_h_i_f_t -15
     ar  s1                ; + tabeladdr;

     ga  rb3   , arn rb4   ; rb3 := Raddr; R:= _3_9 0 _1 1;
b2:  ck  _0                 ; R := R _s_h_i_f_t rb2;
b3:  ab  _0                 ; R := R ∨ tabel[rb3];
     gr (rb3)              ; tabel[rb3] := R;
     hr  s3                ; return
b1:  qq 31.39
b4:  qq 1.39
_e
;
;       _p_r_o_c_e_d_u_r_e spejling
;       qq addr m
;
;  Fra Sørens kode
;
_b b10
                  ; 0000000000 0000000000 0000000012 3456789abc start
                  ; 0000000000 0000000000 0000000021 34a9bc6578 facit
c2:  grn rb1   , grn rb    ; spejlingssekvens
     arn (s1)  , tk  22    ; 0000000000 0000000000 0000000012 3456789abc
                           ; 0000001234 56789abc00 0000000000 0000000000
     ga  rb1   , tk  10    ; 56789abc00 0000000000 0000000000 0000000000
     ck  -6    , ga  rb    ; 0000005678 9abc000000 0000000000 0000000000
     tk  10    , ck  -2    ; 9abc000000 0000000000 0000000000 0000000000
                           ; 009abc0000 0000000000 0000000000 0000000000
     ab  rb    , ck  -8    ; 009abc5678 0000000000 0000000000 0000000000
                           ; 0000000000 9abc567800 0000000000 0000000000
     ab  rb1   , tk  5     ; 0000001234 9abc567800 0000000000 0000000000
                           ; 012349abc5 6780000000 0000000000 0000000000
     gr  rb    , mb rb2    ; 002000a000 6000000000 0000000000 0000000000
     sc  rb    , tk  1     ; 02000a0006 0000000000 0000000000 0000000000
     gr  rb1   , arn rb    ; 0103490bc5 0780000000 0000000000 0000000000
     mb  rb3   , sc  rb    ; 0100090005 0000000000 0000000000 0000000000
     tk  -1    , ab rb1    ; 0010009000 5000000000 0000000000 0000000000
                           ; 02100a9006 5000000000 0000000000 0000000000
     ab  rb    , tk -27    ; 02134a9bc6 5780000000 0000000000 0000000000
                           ; 0000000000 0000000000 0000000021 34a9bc6578
     gr (s1)   , hr s2
b2:  qq 136 t512           ; 0010001000 1000000000 0000000000 0000000000
b3:  qq 273                ; 0100010001 0000000000 0000000000 0000000000
b:   qq                    ; 0000005678
                           ; 012349abc5 6780000000
                           ; 0103490bc5 0780000000
                           ; 0003400bc0 0780000000
                           ;
b1:  qq                    ; 0000001234
                           ; 02000a0006
_e
;
;       _p_r_o_c_e_d_u_r_e drejning
;       qq addr m
;
_b b10
c3:  arn(s1)
     ck  -4    , gr  rb1
     tk  -28   , ab  rb1
     mb  rb2   , gr (s1)
     hr  s2
b1:  qq
b2:  udn (p)   , udn (p)
_e
;
;       _p_r_o_c_e_d_u_r_e indsæt
;       qq tabeladdr    s1
;       qq addr m       s2

;
_b b10
c4:  arn(s2)   , gr  rb1   ; m1:=m
     it (s1)   , pa  rb7   ; rb7 := tabeladdr
     pa rb2    t 3         ; init i
                           ; _f_o_r i := 1 _s_t_e_p 1 _u_n_t_i_l 3 _d_o
b3:  arn rb1   , gr  rb4   ; _b_e_g_i_n m2:=m1;
     pa  rb5   t 2         ; init j
                           ; _f_o_r j := 1 _s_t_e_p 1 _u_n_t_i_l 2 _d_o
b6:  hs  rc5               ; _b_e_g_i_n indsæt1(tabel,m2);
b7:  qq
     qq  rb4
     hs  rc2               ; spejling
     qq  rb4
b5:  bt  2     t -1
     hv  rb6               ; _e_n_d j;
     hs  rc3               ; drejning
     qq  rb1
b2:  bt  3     t -1        ; _e_n_d i;
     hv  rb3
     hr  s3
b1:  qq                    ; m1
b4:  qq                    ; m2
_e
;
;       _p_r_o_c_e_d_u_r_e writeinteger(|<p|>,R);
;
_b b20
c7:  gr  rb16  , pp  12
     hv  rb4           NZ
     sy  16    , hr  s1
b4:  pi  512t511           ; OA:=1
     arn rb19  , gr  rb18
b1:  pm  rb16  , dln rb18
     hv  rb2           NZA
     qq                IZA
b2:  gm  rb16  , ck  -10
     ga  rb3
     pa  rb3   t 16    LZ
b3:  sy  0             NOA
     pm  rb18  , dln rb15
     gr  rb18
     pp  p-1   , bs  p
     hv  rb1
     hr  s1
b15: qq  10.39
b16: qq  0                 ; work
b18: qq
;       109876543210
b19: qq 100000000000.39
_e
;
;       _p_r_o_c_e_d_u_r_e find
;
_b b20
c6:  pan rb1   t d1-1      ; _f_o_r i:=0 _s_t_e_p 1 _u_n_t_i_l 127 _d_o
b1:  gr  _0     t 1         ; tabela[i]:=_4_0 0;
     ac  (rb1)         IPA
     hv   rb1          NPA
     grn  rb2              ; m:=0;
b5:  hs   rc1              ; mloop: _i_f findes(tabela,m)
     qq   d1
     qq   rb2              ; _t_h_e_n _g_o_t_o rb6
     hv   rb6              ; _e_l_s_e

     pi  0     t 1015      ; QA:=0;
     arn rb2
     hv  rb8           NZ  ; _i_f m=0 _t_h_e_n
     pi  8     t 1015  LKB ; QA:=KB;
     hv  rb9
b8:  pan rb7   t d2-1      ; i:=-1;
b7:  arn _0     t 1     IPA ; i:=i+1; R:=lovlige[i];
     gr  rb10  , mb rb2    ; v := R; R := v ∧ m;
     sr  rb10              ; R := R-v;
     hv  rb11          NZ  ; _i_f (v∧m)=v _t_h_e_n _b_e_g_i_n
     arn rb10              ; R:= v;
     ab  0 DX              ; R:= -, v;
     mb  rb2   , gr rb12   ; R:= m ∧ (-, v);
     hs  rc1               ; _i_f findes(tabela, m∧-,v)
     qq  d1
     qq  rb12
     hv  rb9               ; _t_h_e_n _g_o_t_o usikker;
b11: hv  rb7           NPA ; _e_n_d loop lovlige;
     pi  8 t 1015          ; QA:=_t_r_u_e;
b9:  hv  rb6           NQA ; usikker: _i_f QA _t_h_e_n
     hs  rc4               ; indsæt(tabela,m);
     qq  d1
     qq  rb2
     hv  rb6           LKA ; skip print if KA
     sy  64    , arn rb2   ; writecr; R:=m;
     hs  rc7
b6:  arn rb2   , ar rb3    ; m:=m+1;
     gr  rb2   , sr rb4    ; _i_f m |= 4096 _t_h_e_n
     hv  rb5           NZ  ; _g_o_t_o mloop;
     hr  s1                ; return
b2:  qq                    ; m
b3:  qq 1.39
b4:  qq 4096.39
b10: qq                    ; v
b12: qq                    ; m∧-,v
_e
; Layout for print sequence:
; qq <b>.3+<h>.7+<f1>.9+<d>.13+<n>.14+<bE>.17+<f2>.19+<g1>.23+<g2>.27+<g3>.31+<g4>.35+<g5>.39
; 0 _< b _< 15   No. of significant digits
; 0 _< h _< 15   Digits before .
; 0 _< d _< 15   No. of decimals
; 0 _< f1 _< 3   sign
; 0 _< n _< 1    0 in front
; 0 _< bE _< 7   digits in exponent
; 0 _< f2 _< 3   sign on exponent
; 0 _< gx _< 15  grouping
b8:  qq  4.3+ 4.7+0.13+1.14+ 4.23 ;|<nddd|>

[The following taken from  print  in HELP 3]

[NUMBER PRINT ROUTINE                                            page 1]

_b a51                      ;
                           ;
a42: qq  _0   , ck   0      ; entry address part 0_<x<1024
a50: qq  570  , tl  -30    ; entry address part -512_<x_<511
c9:  pm  28   DV           ; entry integer
     pm  -11  D            ; entry fractional
c8:  pa  ra11 X    t485    ; entry real; numberpart:=_t_r_u_e;
     gs  ra12 , gp  ra13   ;   save p; save s;
     ga  ra14 , gm ra23    ;   store exp2; store numberpart;
     pm  (s)  , arn  s     ;
     pm  (s1)           NA ;   get layout address
a14: psn  _0   X            ;   s:=exp2;

     ps  s11  , cl  -20    ;   s:=exp2+11;
     tk  14   , ga  ra17   ; unpack layout: b
     tk  10   , pa  ra16   ;   exp10:=0;
a40: it  _0    , pa  ra4    ; expprinting:
     pp  256  , ck  -6     ;   bE:=256;
     ga  ra18 , tk  10     ;   h
     ck  -8   , ga  ra19   ;   f1
     tk  10   , ck  -6     ;
     ga  ra20 , tk  11     ;   d
     tk  -20  , gt  ra21   ;   -n
     tk  20   , ck  -1     ;
     ga  ra22 , tk  -6     ;   bE+f2
     ca  1    , pp  10     ;   _i_f bE=1 _t_h_e_n minexp:=10;
     ca  2    , pp  100    ;   _i_f bE=2 _t_h_e_n minexp:=100;
     tln 34   , ar  ra43   ;   group picture
     gr  ra8  , snn ra23   ;   R:=-abs(numberpart);
     pa  ra4  V  t-15   LZ ;   _i_f numberpart=0 _t_h_e_n H:=-15
a49: nk  ra46 XV           ; reconversion:  _e_l_s_e x:=numberpart×2|∧exp2;
     gr  ra26 , hv  ra31   ;   _g_o_t_o _i_f numberpart=0 _t_h_e_n L3
     bs s-11  , hv  ra47   ;    _e_l_s_e _i_f s>0 _t_h_e_n conversion1
                           ;    _e_l_s_e conversion2;
a34: hv  ra48 , sr  ra26   ; round x2: R:=rounded x;
a17: pa   _0   XVD  t11  NO ;  _i_f -,overflow _t_h_e_n b:=11
     mt  ra7  , hv  ra49   ;    _e_l_s_e _g_o_t_o reconversion;
     bs (ra4) , ntn (ra4)  ;   _i_f H>0 _t_h_e_n _b_e_g_i_n R:=0;
a21: qq (ra18)             ;    h:=h-H _e_n_d _e_l_s_e h:=h-n;
a19: pp   _0   , gt  ra13   ;   p:=f1; a13incr:=R;
     bs  p509 , hv  ra36   ;   _i_f p|=3 _t_h_e_n _g_o_t_o count h
a11: arn 485  D   t-485    ;   exppart:=-,numberpart; numberpart:=_f_a_l_s_e;
     hs  ra5            LT ;   _i_f exppart _t_h_e_n write _{10}
     bs (ra24), arn ra23   ;   _i_f b1>0∧x<0 _t_h_e_n
[21.9.1967                   (10)                     PRINT, PAIR  page 18]

[NUMBER PRINT ROUTINE                                            page 2]

     arn  -480 DV       NT ;   R:=-
     arn  32   DV          ;    _e_l_s_e R:=+
     bs  p510  , ck  10    ;   _i_f R<0∧p<2 _t_h_e_n R=small
     hs  ra37  , pp  3     ;   write sign
a36:                       ;
a18: bt  _0           t-1   ; count h:
     hsn ra2   , hv  ra36  ;   write space before digits
     bs  p509  , hv  ra11  ;
a13: pp   _0    , ncn  _0    ;   restore p;
     hsn ra1   , qq        ;
     bt (ra4)      t-1     ;   count H
     hsn  ra   , hv  r-1   ;   write digits before point
     arn ra38  , bs (ra20) ;   _i_f d>0 _t_h_e_n
     hs  ra5   , it  -1    ;    write point;
a20: bt  _0     , hh ra39   ;   count d, write decimals
a12: ps  _0     , xrn       ;   restore s; M:=0;
     bs (ra24) , pm ra16   ;   _i_f b2>0 _t_h_e_n M:=exp10;
     can (ra22), hr  s1    ; EXIT
     gm  ra23  , srn  ra9  ;
     pm (ra22) DX          ;
     ps  9     , hh  ra40  ;   _g_o_t_o expprinting;
a37: ca  p     , hh  s     ;   _i_f p=0∧Raddr=0 _t_h_e_n _g_o back
     gr  ra26  , arn ra9   ;   sign not counted in group
     ac  ra8   , arn  ra26 ;
a5:  bs  (ra)  , hv  ra1   ;   _i_f b1>0 _t_h_e_n write digit
     mb  ra44  , ga  ra6   ;
a39: hvn ra10  , arn ra3   ;   _e_l_s_e _i_f digit|=0 _t_h_e_n write digit
a4:  it  _0          t1     ;   count H                        _e_l_s_e write 0
     bs  0     , hvn  ra5  ;   _i_f H<0 _t_h_e_n write 0
a:   bt  _0         t-1     ;   count b1
     mln ra3   , tk  30    ;   next digit in R
a1:  ar  16    D        LZ ;   zero instead of space
a2:  ga  ra6   , bs (ra50) ;   _i_f actual case=upper _t_h_e_n
     mt  ra7   , it  510   ;     R:=-R;
     sy (ra50)   t-510  LT ;   write case;
a10: qq (ra42)      t1     ;   count printed;
a51: arn ra8   , sr  ra9   ;   actual group:= actual group-1;
     gr  ra8   , nc  -273  ;   _i_f -,group full _t_h_e_n write out
a6:  sy  _0     , hhn  s    ;    _e_l_s_e _i_f out>58 _t_h_e_n _b_e_g_i_n
     tk  4     , it  58    ;    actual group:= next group;
     bs (ra6)  , hv r1a51  ;   write out _e_n_d
[21.9.1967                   (10)                     PRINT, PAIR  page 19]

[NUMBER PRINT ROUTINE                                            page 3]

a44: sy  -256  , ud  ra10  ;    _e_l_s_e _b_e_g_i_n writespace; count printed;
                           ;    actual group:=next group-1; write out _e_n_d;
a41: hh  ra51  , it  1     ; conversion:
a28: qq (ra4)       t-1    ; _b_e_g_i_n _c_o_m_m_e_n_t      (count H)
a25: ps  s-3   , nk  r1    ;   by multiplication by 2|∧3/10 or 10/2|∧4
     ps  s_0    , gr  ra26  ;   x is converted to form
     pm  ra26  , bs (ra46) ;   x=x2×10|∧H where x>x2_>.1
a47: mkn ra27  , hh  ra41  ; conversion 1:
a46: tk  s     , gr  ra26  ;
a48: ps  s7    , ar  ra29  ; conversion 2:
     pm  -1    DV       LT ;
     mkn  ra30 , hv  ra28  ; _e_n_d conversion;
     ps  (ra4) , can (ra22);   s:=H; _i_f bE=0∧f2=0 _t_h_e_n _b_e_g_i_n
     bs  s-15  , it  64    ;   _i_f s>15 _t_h_e_n bE=1 _e_n_d;
a22: ca  _0     , hv  ra31  ;   _i_f bE∧f2=0 _t_h_e_n _g_o_t_o L3;
     arn  ra17 , sr  ra20  ;   R:=b-d;
     ga  ra32  , sr  ra18  ;   a32:=b-d; Raddr:=b-d-h-1;
a7:  mb  -1    DX       LT ;   M:=_i_f b-d-h-1<0 _t_h_e_n b-d-h-1 _e_l_s_e -1;
a38: xr  315   , it (ra4)  ; L1: R:=M;
a32: bs  _0     , hs  ra33  ;   _i_f b-d>H _t_h_e_n change exp10
     mt  ra7   , it (ra18) ; L2: R:=-R;
     bs (ra4)  , hs  ra33  ;   _i_f H>b _t_h_e_n change exp10
a31: arn ra20  , ga  ra    ; L3: b1:=d;
a45: ar  ra4   , ga  ra24  ; L4: b2:=H+d;
a24: bs  _0     , ga  ra    ;   _i_f b2>0 _t_h_e_n b1:=b2;
     it (ra17) , bs  (ra24);   _i_f b<H-d _t_h_e_n
     arn ra17  , hh  ra45  ;    b2:=b; _g_o_t_o L4;
     arn  256  D        NT ;   _i_f b2_>0 _t_h_e_n R:=.5;
a35: ps  (ra24), pm  ra21  ; rounding: s:=b2;
     bs  s511  , hh  ra34  ;   _i_f s_<0 _t_h_e_n _g_o_t_o round x2;
a16: xr  p_0    , mln ra29  ;   R:=R×.1;
     ps  s-1   , hh  ra35  ;   s:=s-1; _g_o_t_o rounding;
                           ; change exp10:
a33: ac  ra16  , bs (ra16) ;   exp10:=exp10+R; _i_f exp10_>minexp _t_h_e_n
     sc  ra4   , hh  s-1   ;    _b_e_g_i_n H:=H-R; _g_o_t_o L1∨L2 _e_n_d;
     sc  ra16  , hv  ra31  ;   _e_l_s_e exp10:=exp10-R; _g_o_t_o L3;
                           ;
a3:  qq   10.39            ;
a9:  qq   1                ;
a27: can  s409 , cm (r-410);   0.8
a29: vy   p51  , mln (204) ;   0.1
a30: qq   320              ;   10/16
a43: qq  -17.5+1.25-1.39   ;
a8: qq
a26: qq
a23: qq
_e                          ; _e_n_d write;

_b b10
;
;       _p_r_o_c_e_d_u_r_e findlovlige;
;
_b b20,e20
;
;  findlovlig(m)
;  qq addr m
;  retur fundet
;  retur ikke fundet
;
e1:  pa rb2    t d2-1      ; i:=-1;
b2:  arn _0     t 1     IPA ;loop: i:=i+1; R:=liste[i]; PA:=marks liste[i];
     sr  (s1)               ; _i_f R=m _t_h_e_n
     hr  s2             LZ  ;    return s+2
     hv  rb2            NPA ; _i_f -,PA _t_h_e_n _g_o_t_o loop;
     hr  s3                 ; return s+3
;
;  tilføj lovlig(m)
;  qq addr m
;
e2:  arn d2            IPA ; _i_f liste[0]<0 _t_h_e_n _b_e_g_i_n
     hv  rb4           NT  ;
     arn (s1)              ; R:=m;
     gr  d2            MA  ; liste[0]:=R Amarked;
     hv  rb20              ; _g_o_t_o print _e_n_d _e_l_s_e
b4:  hs  re1               ; _b_e_g_i_n  findlovlig(m);
     qq  (s1)
     hr   s2               ; findes allerede: return s2;
     acn (b2)          M   ; fjern mærkebits fra sidste entry
     arn (s1)              ; R:=m;
     is (b2)
     gr  s1            MA  ; liste[i+1]:=R Amarked;
b20: hr  s2            LKA ;print: skip print if KA
     sy  64                ; writecr;
     hs  rc7               ; writeinteger(|<p|>,m);
     hr  s2                ; return
c10: arn rb3               ; _p_r_o_c_e_d_u_r_e findlovlige; R:=-1;
     gr  rd2           MA  ; lovlige[0] := -1 Amarked;
     pa  rb5   t d8-1      ;
b5:  arn d8-1  t 1     IPB ;_f_o_r row:=1 _s_t_e_p 1 _u_n_t_i_l Bmarked _d_o
     is (rb5)  , is (s)
     it  s-1   , pa  rb6   ; b6 := addr first col in row row;
b6:  arn _0     t1      IQA ; _f_o_r col1:=1 _s_t_e_p 1 _u_n_t_i_l Amarked _d_o
     is  (rb6) , it s-1
     pa   rb7              ; b7:=col1-1;
     grn  rb8              ; m:=0;
b7:  arn _0     t 1     IRA ; _f_o_r col2:=col1 _s_t_e_p 1 _u_n_t_i_l Amarked _d_o
     ab  rb8   , gr  rb8   ; m:=m ∨ tabel[row,col2];
     gr  rb9               ; m1:=m
     pa  rb11  t 3
b15: pa  rb10  t 2         ; _f_o_r i:=1 _s_t_e_p 1 _u_n_t_i_l 3 _d_o _b_e_g_i_n
     arn rb9   , gr rb13   ; _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l 2 _d_o _b_e_g_i_n m2:=m1;
b14: hs  re2
     qq  rb13              ; tilføj lovlig(m2);
     hs  rc2
     qq  rb13              ; m2:=spejling(m2);
b10: bt  2     t -1
     hv  rb14              ; _e_n_d j;
     hs  rc3               ; m1:=drejning(m1);
     qq  rb9
b11: bt  3     t -1
     hv  rb15              ; _e_n_d i
     hv  rb7           NRA ; _e_n_d col2

     hv  rb6           NQA ; _e_n_d col1
     hv  rb5           NPB ; _e_n_d row;
     hr  s1                ; return

b3:  qq -1.39
b8:  qq                    ; m
b9:  qq                    ; m1
b13: qq                    ; m2
_e; findlovlige
;
;       main
;
_b b20
c0:  zq                    ; stop
     vy  32                ; select(32);
     zl                    ; clock:=0.0;
     hs  rc10              ; findlovlige;
     zl        , grf rb2   ; b2:=clock;
     sy  49    , sy 64     ; writetext(|<<a|>); writecr;
     vy  16    , sy 64     ; select(16); writecr;
     qq  rb1   , hs rc8    ; write(|<nddddddddddd.dd|>,RF);
b3:  zq                    ;loop: stop
     vy  32    , zl        ; select(32); clock:=0.0;
     hs  rc6               ; find;
     zl        , grf rb2   ; b2:=clock;
     sy  49    , sy 64     ; writetext(|<<a|); writecr;
     vy  16    , sy 64     ; select(16); writecr;
     qq rb1    , hs rc8    ; write(|<nddddddddddd.dd|>,RF);
     hv  rb3               ; _g_o_t_o loop;
b1:  qq 14.3+12.7+2.13+1.14+12.23+ 2.27 ;|<nddddddddddd.dd|>
b2:  qq
_e ;main
_ec0